1 A short manual of FORM

Symbolic Manipulation for the expansion of Feynman diagrams can easiest be done with FORM. The FORM program can bbe obtained from the form site. Most likely the 32-bits binary should work without problems. The standard way to install it is to have a directory named bin in your home directory and put form in that directory. Usually there is a shell variable named path which will contain this directory and this means that if you type in

form filename

the shell will find the form program in this bin directory and execute it. If your path variable doesn't contain this bin directory you have to edit the file .cshrc or .bashrc (depending on which shell you use) and add the line

```
set path = ( . ~/bin $path )
```

In the bash shell it may be sufficient to have the line

```
set PATH = .; "/bin; $PATH
```

When you pick up FORM and put it in your bin directory make sure that its execution flag has been set. This means that when you do an ls -l in that directory you should see something like

```
-rwxr-xr-x 1 user group 1331128 2007-10-21 17:55 form
```

If the x characters in the first part are missing you need to type the command

```
chmod 755 form
```

This make form 'executable'. After this you can use FORM from any directory.

One prepares a file with a form program of which the extension should be .frm, eg test.frm. This program is then executed with

```
form test
```

which produces output on the screen. In case it is executed with

```
form -f test
```

it produces both output on the screen and (more completely) in the file test.log. This file can then be put in ones favorite editor. FORM can also produce other files, depending on the program.

FORM programs consist of modules. Each module again consists of declarations, definitions and statements. Each module is compiled when its turn has come and then executed. Hence at the level of modules FORM acts as an interpreter, at the level of individual statements as a compiler. This gives a high degree of flexibility as will be seen.

Modules are ended with instructions of which the name starts with a period. The most important instructions are for our purposes:

- .end Executes the current module and terminates the program afterwards.
- .sort Executes the current module, clears the buffers and continues with the next module.

Other module instructions will not play a role in this course.

1.1 General

The names of commands and built in variables are case insensitive. The built in variables all have a name that ends in an underscore. User defined variables are case sensitive and are not allowed to contain the underscore character. There is one exception to this last rule. User defined variables can have any characters in their name provided the name starts with the character valid name for a variable. It is different from [A + B]. The built in imaginary variable i is given by i_{-} which is the same as I_{-} . All statements inside the modules should be terminated with a semicolon.

1.2 Modules

The contents of each module can consist of any of the following in the given order:

- Declarations
- Settings
- Definitions
- Executable Statements
- Output control

None of these has to be present but if more than one type occurs the relative order has to be as specified above. Generically they are all called statements as well.

1.3 Declarations

The important declarations are:

- Symbol Declares the given objects as regular commuting variables.
- Vector Declares the given objects as vectors.
- Index Declares the given objects as indices (of vectors, functions or tensors)
- CFunction Declares the given objects as commuting functions.
- Function Declares the given objects as non-commuting functions.
- Tensor Declares the given object to be a (commuting) tensor. Tensors can have only indices or vectors for their arguments.

In addition one can put the word AutoDeclare in front of such a declaration. This means that all variables of which the name starts with the characters given and is not otherwise declared will become of the mentioned type. More than one variable can be declared at the same time and the notation $i1, \dots, i12$ is understood as the sequence of the twelfe variables i1 till i12.

Vectors either have a single index between parentheses or are part of a dotproduct. Functions have zero or more arguments. The arguments can be empty. An empty argument is interpreted as zero. If a tensor has a vector for an argument it is interpreted that there was in index there that has been contracted with the index of the vector. Indices can have customized dimensions, but we will not look at that here. The default dimension for indices is four.

1.4 Settings

There are not many settings that we will look at. The most important in this course is the Format statement.

```
Format C;
```

will format potential output in such a way that a C compiler can translate the formulas.

```
Format Fortran;
```

will format potential output in such a way that a Fortran compiler can translate the formulas.

```
Format;
```

will switch back to the default FORM mode.

1.5 Definitions

A definition defines an expression as an object for symbolic manipulation. There are several types of expressions, but we will only consider the local expressions. They are defined as in

```
Local F = (a+b)^4;
```

This defines the expression F and its initial value is $(a+b)^4$. The first thing FORM will do is work out the power as $a^4 + 4a^3b + \cdots$. One is allowed to use previously defined expressions in the right hand side of a definition. For reasons of efficiency it is best that there is a .sort instruction before such a previously defined expression is being used.

Expressions are the objects we are interested in. They consist of terms like $4a^3b$. Statements act on the terms of an expression. At the end of the module the results of these manipulations are brought together and sorted into a 'normal ordered' object again.

There are two statements that have the same status as a definition and should be used together with the definitions before the regular statements. They are:

```
Drop [names of expressions];
```

and

```
Skip [names of expressions];
```

The Drop statement removes expressions from the system. The mentioned expressions can still be used in the right hand side of expressions or statements in the current module. After the current module there is no more memory of the given expression(s). If no names of expressions are given all currently defined expresseions will be dropped. This can be done as in

```
Drop;
Local M = Amp*AmpC;
```

The expressions Amp and AmpC will be dropped and a new expression that is the product of the two will be created.

The skip command just instructs FORM that the mentioned expressions are not to be treated by the statements in the current module. It is a way to selectively act on some expressions while not action on others. If no expressions are mentioned all currently active expressions will be skipped as in

```
Skip;
Local AmpC = Amp;
```

where all expressions except for AmpC will be skipped. The nskip statement does the opposite:

```
Skip;
Nskip Amp;
```

will skip all expressions except for the expression Amp.

1.6 Executable Statements

The statements come in the largest variety. We will see however only a small number of them. They act on the individual terms of all expressions The most important statement is the id or identify statement as in

```
id pattern = expression;
```

In this case the pattern will be looked for in the terms of the active expressions and replaced by the right hand side expression as in:

```
id a = b+c;
```

Notice that the pattern is matched and taken out as many times as possible and only after that the right hand side is inseterd as many times as the pattern fit. Hence this will cause no problems with a statement like

```
id a = a+1:
```

If repeated action is needed one should put the statement(s) inside a repeat loop:

```
repeat;
   id a^2 = a+1;
endrepeat;
```

Generic variables or wildcards in the pattern are indicated by a trailing questionmark. In the right hand side the questionmark is not needed:

```
id a^n? = a^(n+1)/(n+1);
```

A very special type of wildcarding is done with the arguments of functions. These wildcards stand for an unspecified number of arguments of the function. Example:

```
repeat;
id f(i1?,i2?,?a)*f(i2?,i3?,?b) = f(i1,i3,?a,?b);
endrepeat;
```

Here the 'argument field' wildcards are given by a questionmark followed by a name. They will match with any number of arguments, also zero. Note also that because i_2 occurs twice in the pattern both occurrences have to be identical. The above is a way of stringing matrices together.

Another useful statement is the trace statement which comes in two varieties. We will use only the trace in 4 dimensions which is given by the trace4 statement. The statement

```
Trace4,i;
```

will take the 4-dimensional trace of all gamma matrices that are marked as belonging to the fermion line i. I is either an index or a (short) positive number.

The multiply statement multiplies all terms with the expression given in the statement:

```
Multiply,right,replace_(U,UB,UB,U,V,VB,VB,V);
```

This multiplies on the right. It is also possible to multiply on the left and if there is no keyword the system either multiplies on the right or on the left, whatever it likes. The replace_ function is described in the subsection on special functions.

There is also an if-statement in which questions are asked for each term and depending on the answer the statements between the if-statement and the corresponding endif statement will be executed. For more details we refer to the complete FORM manual.

1.7 Output control

Here we mention for instance whether we want the output of the current module to be printed. We can also specify whether we want a level of brackets in the output. The print statement has some varieties:

```
Print;
Print +f;
Print +s;
Print expressionnames;
```

In the first case all active expressions (that are not dropped or skipped) will be printed. In the second case all will be printed but if FORM was called with the -l flag they will be printed only in the log file. In the third case all will be printed with only a single term per line. Finally in the fourth case only the mentioned expressions will be printed. All these options can be combined. In that case the +f and/or +s come before the name(s) of the expression(s).

The bracket statement specifies whether some objects should be taken outside brackets when the output is printed. This can make the output easier to read. Each bracket is started on a new line.

```
Bracket x,y,f;
```

In this case (x and y symbols, f a function) all powers of x and y and all occurrences of f are placed outside brackets. Each different occurrence of powers of x and y and the function f defines a new bracket. Note that this has also an influence on the ordering of the terms because now powers of x which go outside the brackets are more significant than powers of a which will be inside the brackets. Without bracket statement this might be different depending on whether a is declared before x.

1.8 Special Functions

Amoung the various special functions in FORM are the Dirac gamma matrices and a number of delta functions. There is also the Levi-Civita tensor and much more. We will only need the ones mentioned here.

The Dirac gamma matrices are indicated by g_. They are noncommuting objects that have in principle two arguments: The first argument is either an index or a (small) positive integer that indicates to which spinline the matrix belongs. Matrices of different spinlines will commute with each other. The second argument is either an index or a vector. If it is a vector it is a shorthand

notation for an index that is contracted with the index of the vector. There are several special indices for the second argument: 5_{-} indicates γ_5 while 6_{-} indicates $(1 + \gamma_5)$ and 7_{-} stands for $(1 - \gamma_5)$. If the second argument is absent we have the unit matrix. We have several shorthand notations:

```
gi_{(j)} = g_{(j)}

g5_{(j)} = g_{(j,5_{(j)})}

g6_{(j)} = g_{(j,6_{(j)})} = gi_{(j)}+g5_{(j)}

g7_{(j)} = g_{(j,7_{(j)})} = gi_{(j)}-g5_{(j)}
```

A string of gamma matrices can be written together as in

```
g_{j,nu,p1,mu,p2} = g_{j,nu}*g_{j,p1}*g_{j,mu}*g_{j,p2}
```

The Levi-Civita tensor \mathbf{e}_{-} is a totally antisymmetric tensor. In four dimensions we have $e^{0123} = 1$. It has close relations with $gamma_5$:

```
Indices mu,nu,ro,si;
Local F = g5_(j)*g_(j,mu)*g_(j,nu)*g_(j,ro,)*g_(jsi);
Trace4,j;
Print;
.end
    F = e_(mu,nu,ro,si);
```

For this course we need to know only two delta functions. The Dirac delta function $\delta^{\mu\nu}$ and the replace_ function. The Dirac delta is the function \mathbf{d}_{-} with two index arguments. As repeated indices are automatically summed over most \mathbf{d}_{-} functions disappear quickly and have the effect of renaming indices. Something similar is achieved for other objects with the replace_ function. It should have an even number of arguments. The arguments come in pairs and when a term has a replace function its effect is that in the term everywhere the first element of the pair is replaced by the second element. Example

```
Multiply replace_(U,UB,UB,U,V,VB,VB,V);
```

Each term will be multiplied by this replace function after which in the term U is replaced by UB, UB by U, V by VB and VB by V. Then the replace function is removed as it has done its work. This works much faster and less complicated than trying to do this with a number of id-statements.

1.9 The Preprocessor

Just like the C compiler FORM has a rather powerful preprocessor with a whole range of preprocessor instructions and its own preprocessor variables. Preprocessor instructions start with the character # as in

```
#include amplitude.h
```

which would be replaced by the contents of the file amplitude.h. The preprocessor has its own variables. When they are used they are between a matching set of backquote and quote. When this is encountered a textual replacement is made and compilation continues at the beginning of the content of the variable as in

```
#define MAX "20"
#do i = 1,'MAX'
    id a'i' = a{'i'+'MAX'};
#enddo
```

In the define instruction MAX is defined as the string 20. Preprocessor variables contain always character strings unless the preprocessor calculator is invoked in which case an attempt is made to interpret the string as an arithmatic expression.

In the do instruction 'MAX' is replaced by the string 20and the program will go through the loop twenty times, each time generating an instruction. The first time i is the string 1 etc. The curly brackets in the right hand side invoke the preprocessor calculator and hence the first time {1+20} is evaluated and replaced by the string 21 and hence we generate in total twenty statements of the type

```
id a1 = a21;
id a2 = a22;
...
id a20 = a40;
```

Another preprocessor construction is the procedure. each procedure can be part of the regular input stream and is then read into memory and kept there, or it resides in a separate file which has the name of the procedure and the extension .prc but in our examples we have put the procedure squareamplitude in the header file amplitude.h. procedures can have arguments and FORM is very careful with the syntax of the procedure instruction and the corresponding call instruction that invokes the procedure. The procedure instruction is given by

```
#procedure name(argument(s))
     contents
```

#endprocedure

The variables in the arguments are preprocessor variables and hence when used should be enclosed in a matching pair of backquote and quote. It is not necessary to have arguments. Because the procedure is handled entirely by the preprocessor it is actually a gigantic macro. The procedures are called by the call instruction as in

```
#call squareamplitude(Amp,Mat)
```

in which case the instruction is replaced by the contents of squareamplitude in which the first argument is replaced by the string Amp and the second by the string Mat. Note that procedures can call other procedures and even recursions are possible here provided that one takes care that the recursion terminates.

Finally one more feature of the preprocessor to make the creation of complicated input easier:

```
\label{eq:multiply replace_(i1,i21>,...,i20,i40>);} \\
```

The pair <> is used here as a type of bracket to define a pattern and the operator indicated by three dots indicates that we talk about a range. This notation needs a starting pattern and a finishing pattern and then the preprocessor will generate the whole range, in this case a replace_ function with 40 arguments.

1.10 Dollar variables

Apart from regular algebraic variables and the string valued preprocessor variables FORM knows a third type of variables that can be accessed both by the preprocessor and the algebraic execution unit. These variables have a name that starts with a \$ sign. These variables can contain numbers or small algebraic expressions. When referred to between a matching pair of a backquote and a quote, their value is translated into a string and used as a preprocessor variable. Without the backquote and quote their value is substituted during algebraic execution at the term level. These variables are given a value by the preprocessor if preceded by the character # and during algebraic execution without this character. Example:

```
#$count = 0;
$count = $count + 1;
.sort
#do i = 1,'$count'
etc
```

During compilation the variable \$count gets the value zero. Then during execution each time the program passes here (for each term) its value is raised by one. After the .sort it will contain the number of terms that FORM had before the sort. This number is then used by the preprocessor as a parameter in the do loop. There are more ways to give values to the dollar variables and one can do very complicated things with them. This is however outside the scope of this small manual.

2 Calculating Matrix elements

There are many ways to calculate Matrix elements from amplitudes. Here we will look at one way based on the use of FORM and the procedure squareamplitude in the header file amplitude.h. We look at the amplitude for the calculation of the reaction $e^-e^+ \to \tau^-\tau^+$ followed by the decays $\tau^- \to e^-\overline{\nu}_e\nu_\tau$ and $\tau^+ \to \mu^+\nu_\mu\overline{\nu}_\tau$. The FORM program based on amplitude.h is

```
#-
#:ContinuationLines 100

*
* The matrix element for the reaction
* e-e+ -> tau-tau+ -> nu-tau e- nubar-e nu-mu mu+ nubar-tau
* At low energies there is only one diagram!

*
#include amplitude.h
*
Local Amp = VB(i1,pb,me)*g(i1,i2,j1)*U(i2,pa,me)
    *phprop(j1,j2,q)
    *UB(i3,p1,mnt)*g(i3,i4,j3)*g(i4,i5,k7)
    *fprop(i5,i6,q1,mt)*g(i6,i7,j2)*fprop(i7,i8,-q2,mt)
    *g(i8,i9,j4)*g(i9,i10,k7)*V(i10,p6,mnt)
    *UB(i11,p2,me)*g(i11,i12,j3)*g(i12,i13,k7)*V(i13,p3,mne)
    *UB(i14,p4,mnm)*g(i14,i15,j4)*g(i15,i16,k7)*V(i16,p5,mm)
    ;
#call squareamplitude(Amp,value)
.sort
```

```
Symbols widthtau, s, factor;
 Format C;
 * In the narrow width expansion we can replace
      the tau propagators:
 id prop(q1.q1-mt^2)^2 = 1/2/mt/widthtau;
 id prop(q2.q2-mt^2)^2 = 1/2/mt/widthtau;
 id prop(q.q) = 1/s;
 id 1/s^2/widthtau^2 = 4*factor*mt^2;
 id p1.p1 = mt^2;
 id p2.p2 = mt^2;
 id mt^2 = mt2;
 id me^2 = me2;
 id mt2^2 = mt4;
 Bracket factor;
 Print +f;
  .sort
 #$fac = factorin_(value);
 Multiply 1/('$fac');
 Print +f;
  .sort
 Skip;
 #write <> " value = value * (%$);",$fac
  .end
and the file amplitude.h is given by
 AutoDeclare Index i,j,k;
 AutoDeclare Symbol m,x;
 AutoDeclare Vector p,q;
 Vector q,pe,pp,pa,pb,q1,q2,p1,...,p10;
 CF UB,U,VB,V,g,gstring,e;
 CF fprop,phprop,gprop,prop;
 #procedure squareamplitude(Amp,Mat)
  .sort
     We skip everything but Amp. In Amp we look for the highest
      i and j indices
 Skip;
 NSkip 'Amp';
 #$imax = 0;
 \#do i = 1,40
      if ( match(VB(i'i',?a)) || match(V(i'i',?a))
           || match(UB(i'i',?a)) || match(U(i'i',?a))
           || match(g(i'i',?a)) || match(g(i?,i'i',?a))
```

```
|| match(fprop(i'i',?a))
         || match(fprop(i?,i'i',?a)) );
            $imax = 'i';
    endif;
#enddo
#$jmax = 0;
#do j = 1,20
    if ( match(g(?a,j'j')) ||
         match(phprop(j'j',?a)) ||
        match(phprop(j?,j'j',?a)));
        jmax = 'j';
    endif;
#enddo
.sort
    Just for a check we print the highest i and j indices
#message highest i is i'$imax', highest j is j'$jmax';
    Now construct the conjugate
Skip;
    'Amp'C = 'Amp';
id i_{-} = -i_{-};
    Make a new set of dummy indices above $imax and $jmax.
Multiply replace_(
           <i1,i{'$imax'+1}>,...,<i'$imax',i{2*'$imax'}>);
Multiply replace_(
           <j1,j{'$jmax'+1}>,...,<j'$jmax',j{2*'$jmax'}>);
    Exchange rows and columns
id g(i1?,i2?,j?) = g(i2,i1,j);
id fprop(i1?,i2?,?a) = fprop(i2,i1,?a);
id phprop(j1?,j2?,p?) = phprop(j2,j1,p);
    and exchange U and UB, V and VBAR
Multiply replace_(UB,U,U,UB,VB,V,V,VB);
    gamma5 gets a minus sign. Hence k6 <--> k7
Multiply replace_(k6,k7,k7,k6);
id g(?a,k5) = -g(?a,k5);
.sort
```

```
Now multiply Amp and AmpC to get the matrix element squared.
Skip;
Drop, 'Amp', 'Amp'C;
    'Mat' = 'Amp'*'Amp'C;
    Spin sums
   U(i1?,p?,m?)*UB(i2?,p?,m?) = g(i1,i2,p)+g(i1,i2)*m;
   V(i1?,p?,m?)*VB(i2?,p?,m?) = -g(i1,i2,p)+g(i1,i2)*m;
   e(j1?,p?)*e(j2?,p?) = -d_(j1,j2);
    Propagators
id fprop(i1?,i2?,p?,m?) =
                          (g(i1,i2,p)+g(i1,i2)*m)*prop(p.p-m^2);
    phprop(j1?, j2?, q?) = -d_(j1, j2)*prop(q.q);
    String the gamma matrices together in traces.
repeat id g(i1?,i2?,?a)*g(i2?,i3?,?b) = g(i1,i3,?a,?b);
.sort
Skip;
NSkip 'Mat';
    Now put the traces one by one in
    terms of the built in gammas
\#do i = 1,10
    id, once, g(i1?, i1?, ?a) = g_{-}('i', ?a);
    id g_{(i',k7)} = g_{(i')};
    id g_{(i',k6)} = g_{(i')};
    id g_{(i',k5)} = g_{(i')};
#enddo
.sort
    Finally take the traces
\#do i = 1,10
Trace4, 'i';
#enddo
#endprocedure
```

Let us discuss the various aspects of the program.

We define the gamma matrix with its two matrix indices as g(i1,i2,j1) in which i1 and i2 are the spinor indices. The spinor indices in the amplitude should be called i1,i2,etc. The Lorenz indices should be called j1,j2,etc. The spinors are U, UB (for \overline{u}), V and VB (for \overline{v}). Their first argument is the spinor index, the second argument is the fourvector and the third argument should be the mass.

If we define pa as the momentum of the incoming electron and pp the momentum of the incoming positron the trace on the incoming side is

```
VB(i1,pb,me)*g(i1,i2,j1)*U(i2,pa,me)
```

with j1 the Lorenz index of the s-channel photon. The photon propagator is defined as phprop(j1, j2,q). Next comes the tau trace:

```
*UB(i3,p1,mnt)*g(i3,i4,j3)*g(i4,i5,k7)
*fprop(i5,i6,q1,mt)*g(i6,i7,j2)*fprop(i7,i8,-q2,mt)
*g(i8,i9,j4)*g(i9,i10,k7)*V(i10,p6,mnt)
```

in which fprop is a fermion propagator. It has two spinor indices, a momentum and a mass. The variable mt is the mass of the tau and mnt is the mass of the tau neutrino. The variable k7 takes the role of g7_ in due time. The labelling of the momenta is

```
1. p_1: The \tau^-.
   2. p_2: The \tau^+.
   3. p_3: The outgoing \nu_{\tau}.
   4. p_4: The outgoing electron.
   5. p_5: The outgoing \overline{\nu}_e.
   6. p_6: The outgoing \nu_{\mu}.
   7. p_7: The outgoing \mu^+.
   8. p_8: The outgoing \overline{\nu}_{\tau}.
This leaves two traces for the electron and the \mu^+:
```

```
*UB(i11,p2,me)*g(i11,i12,j3)*g(i12,i13,k7)*V(i13,p3,mne)
```

and

```
*UB(i14,p4,mnm)*g(i14,i15,j4)*g(i15,i16,k7)*V(i16,p5,mm)
```

in which mne is the mass of the ν_e and mnm is the mass of the ν_u . The mass of the electron is me and the mass of the muon is mm. This gives the rules for feeding in the amplitudes which are translations of the Feynman diagrams.

Next we look inside amplitude.h. For some declarations we have used autodeclare because we don't know in advance how many of these variables we will encounter in the main program. After the declarations we define the procedure squareamplitude. The first thing this procedure does is to determine the maximum spinor index and the maximum Lorenz index. We assume that there will be no more than 40 spinor indices and no more than 20 Lorenz indices. To exceed these limits would require some really big diagrams. This is needed to avoid that contracted indices in the conjugate of the amplitude will be identical to indices in the original amplitude. These maximum values are stored in the dollar variables \$imax and \$jmax.

Then we define the conjugate of the amplitude and decide to only operate on it (by skipping everything else). Then we replace the spinor and Lorenz indices by indices that are guaranteed to be different. The next three id-statements conjugate the matrices (ie changes the rows and the

columns). Then the U etc are conjugated. Because γ_5 gets a minus sign under conjugation we have to take care of that sign as well. This finishes the conjugation.

In the next module we construct the square of the matrix element by multiplying Amp and AmpC. At the same time we don't need those two any longer. The substitution of U()*UB() and V()*VB() are the famous spin sums and the substitution of two e-functions with the same momentum does the same for external photons. Next we write out the fermion propagators and the photon propagators and finally we string the gamma matrices together.

In the next module we take each time one such string and as it has now two identical spinor indices we have to take its trace. We write it in terms of the built in gamma function and resolve the k5, k6 and k7 as γ_5 , γ_6 and γ_7 . Each trace will have its own spinline 'i'. Finally we take the traces.

The trace algorithms of FORM are quite sophisticated. It knows identities that involve more than one trace at a time and hence it is best to first write all strings to the built in gamma matrices and then take the trace. You can experiment with this by trying

```
#do i = 1,10
    id,once,g(i1?,i1?,?a) = g_('i',?a);
    id g_('i',k7) = g7_('i');
    id g_('i',k6) = g6_('i');
    id g_('i',k5) = g5_('i');
    Trace4,'i';
#enddo
```

You will get a very messy and very lengthy output with terms that contain products of Levi-Civita tensors. When these tensors are contracted and simplifications are used when needed the answer will eventually be the same but after much more work, both of the programmer and of the computer. Hence it is best to let FORM decide on how to do the traces.

At this point we have an expression with masses, dotproducts and some propagators of the form prop(argument). In the current case they can all be simplified. Finally we introduce some variables that are powers of symbols to avoid too much use of the pow function in the language C. The output of this program (suppression the listing of the input) is

#-0.00 sec Time = Generated terms = 1 Amp Terms in output = 1 Bytes used 396 Time = 0.00 sec Generated terms = 1 Terms in output = Amp 1 Bytes used 396 ~~~highest i is i16, highest j is j4 0.00 sec Time = Generated terms = 1 AmpC Terms in output = 1 Bytes used 396 0.91 sec Time = Generated terms = 4096 value Terms in output = 4096 Bytes used 1257062 1.03 sec Time = Generated terms = 128

	value	Terms in output Bytes used	=	128 36814
Time =	1.04 sec value	Generated terms Terms in output Bytes used		164 27 1616
Time =	1.04 sec value	Generated terms Terms in output Bytes used		27 27 1398

value =

+ factor * ($-524288.*pa_pb*q1_q2*p1_p2*p3_p4*p5_p6*mt2 +$ $524288.*pa_pb*q1_p3*q2_p4*p1_p2*p5_p6*mt2 + 524288.*pa_pb*$ $q1_p4*q2_p3*p1_p2*p5_p6*mt2 + 1048576.*pa_q1*pb_q2*q1_p3*q2_p4 \\$ *p1_p2*p5_p6 + 524288.*pa_q1*pb_q2*p1_p2*p3_p4*p5_p6*mt2 - $\verb|pb_p4*q2_p3*p1_p2*p5_p6*mt2| + 1048576.*pa_q2*pb_q1*q1_p3*q2_p4|$ *p1_p2*p5_p6 + 524288.*pa_q2*pb_q1*p1_p2*p3_p4*p5_p6*mt2 - $\verb|pb_p3*q1_p4*p1_p2*p5_p6*mt2| - 524288.*pa_p3*pb_q2*q1_q1*q2_p4*|$ p1_p2*p5_p6 - 524288.*pa_p3*pb_q2*q1_p4*p1_p2*p5_p6*mt2 + 262144.*pa_p3*pb_p4*q1_q1*q2_q2*p1_p2*p5_p6 + 524288.*pa_p3* pb_p4*q1_q2*p1_p2*p5_p6*mt2 + 262144.*pa_p3*pb_p4*p1_p2*p5_p6* $\mathtt{mt4} - 524288.*\mathtt{pa}_\mathtt{p}4*\mathtt{pb}_\mathtt{q}1*\mathtt{q}1\mathtt{_p}3*\mathtt{q}2\mathtt{_q}2*\mathtt{p}1\mathtt{_p}2*\mathtt{p}5\mathtt{_p}6 - 524288.*$ pa_p4*pb_q1*q2_p3*p1_p2*p5_p6*mt2 + 262144.*pa_p4*pb_p3*q1_q1* $q2_q2*p1_p2*p5_p6 + 524288.*pa_p4*pb_p3*q1_q2*p1_p2*p5_p6*mt2$ p1_p2*p3_p4*p5_p6*me2 - 524288.*q1_q1*q2_p3*q2_p4*p1_p2*p5_p6* me2 + 1048576.*q1_q2*q1_p3*q2_p4*p1_p2*p5_p6*me2 - 524288.* q1_p3*q1_p4*q2_q2*p1_p2*p5_p6*me2 + 1048576.*q1_p3*q2_p4*p1_p2 $*p5_p6*mt2*me2 + 262144.*p1_p2*p3_p4*p5_p6*me2*mt4);$

Time = 1.04 sec Generated terms = 27 value Terms in output = 27 Bytes used = 952

value =

 $2*pa_pb*q1_q2*p3_p4*mt2 - 2*pa_pb*q1_p3*q2_p4*mt2 - 2*pa_pb*q1_p4*q2_p3*mt2 - 4*pa_q1*pb_q2*q1_p3*q2_p4 - 2*pa_q1*pb_q2*p3_p4*mt2 + 2*pa_q1*pb_p4*q1_p3*q2_q2 + 2*pa_q1*pb_p4*q2_p3*mt2 - 4*pa_q2*pb_q1*q1_p3*q2_p4 - 2*pa_q2*pb_q1*p3_p4*mt2 + 2*pa_q2*pb_p3*q1_q1*q2_p4 + 2*pa_q2*pb_p3*q1_p4*mt2 + 2*pa_p3*pb_q2*q1_q1*q2_p4 + 2*pa_p3*pb_q2*q1_q1*q2_p4 + 2*pa_p3*pb_q2*q1_p4*mt2 - pa_p3*pb_p4*q1_q1*q2_q2 - 2*pa_p3*pb_p4*q1_q1*q2_q2 - 2*pa_p3*pb_p4*q1_q1*q2_q2 - 2*pa_p3*pb_p4*q1_q1*q2_p3*mt2 - pa_p4*pb_p3*q1_q1*q2_q2 - 2*pa_p4*pb_p3*q1_q2*mt2 - pa_p4*pb_p3*q1_q1*q2_q2 - 2*pa_p4*pb_p3*q1_q2*mt2 - pa_p4*pb_p3*mt4 - q1_q1*q2_q2*p3_p4*me2 + 2*q1_q1*q2_p3*q2_p4*me2 - 4*q1_q2*q1_p3*q2_p4*me2 + 2*q1_p3*q1_p4*q2_q2*me2 - 4*q1_p3*q2_p4*me2 - p3_p4*me2*mt4;$

value = value * (-262144.*p1_p2*p5_p6*factor);
1.04 sec out of 1.15 sec

The execution time was on a pentium 2800.